This worksheet is to help you solve systems of differential equations. We will use the standard health-sickness model and solve the Kolmogorov forward equations. At the end of the worksheet, we will also solve the backward equations. We start bu defining the usual mortality force (this is now the force for healthy lives)

```
> m := t ->
  0.0005+0.00007585775*10^(0.038*t);
```

$$m := t \rightarrow 0.0005 + 0.00007585775\ 10^{(0.038\ t)}$$

We now define the mortality force for sick lives. There are three ways to adjust the mortality upwards and we apply all three of them; you could of course choose to use a totally different function.

```
> ms:= t -> 0.0002+1.5*m(t+5);
```

$$ms := t \rightarrow 0.0002 + 1.5\ \mathrm{m}(t+5)$$

```
> evalf(ms(45));
```

$$0.009988392898$$

We now define the force of transition from the healthy state to the sick state; It is defined as si(x+t) where x is the age of the life. In general it does not have to be of this form, but you have by now been trained to think this way. It is appropriate that it is increasing.

```
> si := t -> 0.1*(1+t/50);
```

$$si := t \rightarrow 0.1\left(1 + \frac{1}{50}t\right)$$

And finally the recovery force from sick to healthy. It is appropriate that it is decreasing. Again we use re(x+t) where x is the age of the life and of course it does not have to be of this form.

```
> re := t -> 2*(1+t/50)^(-1);
```

$$re := t \to \frac{2}{1 + \frac{1}{50}\,t}$$

This is the system of forward equations. The bounary conditions reflect that we assume that the 35 year old life is healthy initially.

```
> dsys1 := {diff(ph(t),t)=re(35+t)*ps(t)-(m(35+t)+si(35+t))*ph(t), diff(ps(t),t)=si(35+t)*ph(t)-(ms(35+t)+re(35+t))*ps(t), ph(0)=1, ps(0)=0};
```

$$dsys1 := \left\{ \mathrm{ph}(0) = 1, \mathrm{ps}(0) = 0, \frac{d}{dt}\mathrm{ph}(t) = \frac{2\,\mathrm{ps}(t)}{\frac{17}{10} + \frac{t}{50}} - (0.1705000000 + 0.00007585775\;10^{(1.330 + 0.038\,t)} + 0.002000000000\,t)\,\mathrm{ph}(t), \frac{d}{dt}\mathrm{ps}(t) = (0.1700000000 + 0.002000000000\,t)\,\mathrm{ph}(t) - \left(0.00095 + 0.000113786625\;10^{(1.520 + 0.038\,t)} + \frac{2}{\frac{17}{10} + \frac{t}{50}}\right)\mathrm{ps}(t) \right\}$$

Suppose we are interested in the various probabilities till time 30.

We solve till time 30;

```
> dsol1 := dsolve(dsys1, numeric, range =
  0..30);
```

$$dsol1 := \mathbf{proc}(x\_rkf45) \ \ldots \ \mathbf{end\ proc}$$

Here are the probabilities at various times that the life is healthy or sick. You can of course get the probability that the life is dead by 1-ph(t)-ps(t)

```
> dsol1(30);
```

$$[t = 30., \mathrm{ph}(t) = 0.584616171884921766, \mathrm{ps}(t) = 0.148467078383623580]$$

```
> dsol1(25);
```

$$[t = 25., \mathrm{ph}(t) = 0.667726051259915132, \mathrm{ps}(t) = 0.156515294135883776]$$

```
> dsol1(20);
```

$$[t = 20., \mathrm{ph}(t) = 0.731880008935261839, \mathrm{ps}(t) = 0.157168895977335132]$$

```
> dsol1(15);
```

$$[t = 15., \mathrm{ph}(t) = 0.781214481184303033, \mathrm{ps}(t) = 0.152679144958058371]$$

```
> dsol1(10);
```

$$[t = 10., \mathrm{ph}(t) = 0.819708200829775912, \mathrm{ps}(t) = 0.144876124041141530]$$

```
> dsol1(5);
```

$$[t = 5., \mathrm{ph}(t) = 0.850713955552217849, \mathrm{ps}(t) = 0.134914070381398076]$$

```
> dsol1(1);
```

$[t = 1., ph(t) = 0.903583164724605647,$
$ps(t) = 0.0940692567402658664]$

Let us plot to discuss any patterns

```
> with(plots);
Warning, the name changecoords has been redefined
```

[*animate*, *animate3d*, *animatecurve*, *arrow*, *changecoords*, *complexplot*, *complexplot3d*, *conformal*, *conformal3d*, *contourplot*, *contourplot3d*, *coordplot*, *coordplot3d*, *cylinderplot*, *densityplot*, *display*, *display3d*, *fieldplot*, *fieldplot3d*, *gradplot*, *gradplot3d*, *graphplot3d*, *implicitplot*, *implicitplot3d*, *inequal*, *interactive*, *listcontplot*, *listcontplot3d*, *listdensityplot*, *listplot*, *listplot3d*, *loglogplot*, *logplot*, *matrixplot*, *odeplot*, *pareto*, *plotcompare*, *pointplot*, *pointplot3d*, *polarplot*, *polygonplot*, *polygonplot3d*, *polyhedra_supported*, *polyhedraplot*, *replot*, *rootlocus*, *semilogplot*, *setoptions*, *setoptions3d*, *spacecurve*, *sparsematrixplot*, *sphereplot*, *surfdata*, *textplot*, *textplot3d*, *tubeplot*]

This is the plot of ph(t)

```
> odeplot(dsol1);
```



We can also plot ps(t)

```
> odeplot(dsol1, [t,ps(t)]);
```



We saw an interesting pattern, with the sickness probability rapidly increasing and then being somewhat stable. Let us extend the time period to 60 years to see what happens. If you are not interested in what happens after time 30, you should not do this as it might affect accuracy. Happily, it does not.

```
> dsol2 := dsolve(dsys1, numeric, range =
  0..60);
```

$dsol2 := \mathbf{proc}(x\_rkf45)\ \ldots\ \mathbf{end\ proc}$

```
> dsol2(30);
```

$[t = 30., \mathrm{ph}(t) = 0.584616150938323088,$
$\mathrm{ps}(t) = 0.148467099946412606]$

```
> dsol2(60);
```

$[t = 60., \mathrm{ph}(t) = 0.00774291047322251628,$
$\mathrm{ps}(t) = 0.00225753826616078604]$

This is the plot of ph(t)

```
> odeplot(dsol2);
```



And this is the plot of ps(t); we now see its behaviour. Can you explain it?

```
> odeplot(dsol2, [t,ps(t)]);
```



We can do more; for example plot survival probabilities.

```
> odeplot(dsol2, [t,ps(t)+ph(t)]);
```



Another interesting quantity is the probability that the life is sick given it is alive. Note that the calculation can not be that accurate for large t.

```
> odeplot(dsol2, [t,ps(t)/(ps(t)+ph(t))]);
```



Let us now assume the life is sick initially.

```
> dsys3 :=
  {diff(ph(t),t)=re(35+t)*ps(t)-(m(35+t)+si(35+t))*ph(t),
  diff(ps(t),t)=si(35+t)*ph(t)-(ms(35+t)+re(35+t))*ps(t), ph(0)=0, ps(0)=1};
```

$$dsys3 := \left\{ \frac{d}{dt}\operatorname{ps}(t) = (0.1700000000 + 0.002000000000\,t)\operatorname{ph}(t) - \left(0.00095 + 0.000113786625\,10^{(1.520 + 0.038\,t)} + \frac{2}{\frac{17}{10} + \frac{t}{50}}\right)\operatorname{ps}(t), \frac{d}{dt}\operatorname{ph}(t) = \frac{2\operatorname{ps}(t)}{\frac{17}{10} + \frac{t}{50}} - (0.1705000000 + 0.00007585775\,10^{(1.330 + 0.038\,t)} + 0.002000000000\,t)\operatorname{ph}(t), \operatorname{ph}(0) = 0, \operatorname{ps}(0) = 1 \right\}$$

```
> dsol3 := dsolve(dsys3, numeric, range = 0..60);
```

$$dsol3 := \mathbf{proc}(x\_rkf45)\ \ldots\ \mathbf{end\ proc}$$

Note that ph(30) and ps(30) are not that different to the ones before. Not so for ph(1) and ps(1)

```
> dsol3(30);
>
```

$$[t = 30., \operatorname{ph}(t) = 0.583417800749720050, \operatorname{ps}(t) = 0.148162764857944996]$$

```
> dsol3(1);
```

$$[t = 1., \operatorname{ph}(t) = 0.641804255159956982, \operatorname{ps}(t) = 0.354383182540079888]$$

Here are the plots. Note that there is a bit of a difference at the beginning, but then the values are close to the ones before. Whether the life was healthy or sick at time 0 matters less and less as time goes on. Try to interpret the ps(t) plot.

```
> odeplot(dsol3);
```




```
> odeplot(dsol3, [t,ps(t)]);
```



And finally the backward equations. We are interested in the probability the life is sick at time 30 under various intial assumptions. ph(t) has a different interpration; it is the probability that the life is sick at time 30 given that it is healthy at time t. Also ps(t) is the probability that the life is sick at time 30 given that it is sick at time t.

```
> dsys4 :=
```

```
{diff(ph(t),t)=-si(35+t)*ps(t)+(m(35+t)+si(
35+t))*ph(t),
diff(ps(t),t)=-re(35+t)*ph(t)+(ms(35+t)+re(
35+t))*ps(t), ph(30)=0, ps(30)=1};
```

$$dsys4 := \left\{ \mathrm{ph}(30) = 0, \mathrm{ps}(30) = 1, \frac{d}{dt}\mathrm{ph}(t) = -(0.1700000000 + 0.002000000000\, t)\, \mathrm{ps}(t) + (0.1705000000 + 0.00007585775\, 10^{(1.330 + 0.038\, t)} + 0.002000000000\, t)\, \mathrm{ph}(t), \frac{d}{dt}\mathrm{ps}(t) = -\frac{2\,\mathrm{ph}(t)}{\frac{17}{10} + \frac{t}{50}} + \left(0.00095 + 0.000113786625\, 10^{(1.520 + 0.038\, t)} + \frac{2}{\frac{17}{10} + \frac{t}{50}}\right)\mathrm{ps}(t) \right\}$$

```
> dsol4 := dsolve(dsys4, numeric, range =
0..30);
```

$$dsol4 := \mathbf{proc}(x\_rkf45)\ \ldots\ \mathbf{end\ proc}$$

Here are some values. Note that some of them are the same with probabilities calculated before as they should be. Note also that it only matters whether the life is hhealthy or sick towards the end.

```
> dsol4(0);
```

$[t = 0., \mathrm{ph}(t) = 0.148467037579746608,$
$\mathrm{ps}(t) = 0.148162731445629886]$

```
> dsol4(25);
```

$[t = 25., \mathrm{ph}(t) = 0.180161977834510545,$
$\mathrm{ps}(t) = 0.179970866022552438]$

```
> dsol4(29);
```

$[t = 29., \mathrm{ph}(t) = 0.134147371106263180,$
$\mathrm{ps}(t) = 0.448247718537325478]$

```
>
```