This worksheet is to demonstrate the solution of Thiele's equations. We will use the standard health-sickness model and investigate a 30 year policy. We are going to assume that a continuouous benefit of 20000 per annum is payable for as long as the life is sick. The continuous premium is payable while he is healthy. We start by defining the usual mortality force (this is now the force for healthy lives)

```
> m := t ->
  0.0005+0.00007585775*10^(0.038*t);
```

$$m := t \rightarrow 0.0005 + 0.00007585775\ 10^{(0.038\ t)}$$

We now define the mortality force for sick lives. There are three ways to adjust the mortality upwards and we apply all three of them; you could of course choose to use a totally different function.

```
> ms:= t -> 0.0002+1.5*m(t+5);
```

$$ms := t \rightarrow 0.0002 + 1.5\ \mathrm{m}(t + 5)$$

```
> evalf(ms(45));
```

$$0.009988392898$$

We now define the force of transition from the healthy state to the sick state; It is defined as si(x+t) where x is the age of the life. In general it does not have to be of this form, but you have by now been trained to think this way. It is appropriate that it is increasing.

```
> si := t -> 0.1*(1+t/50);
```

$$si := t \rightarrow 0.1 \left(1 + \frac{1}{50}\ t\right)$$

And finally the recovery force from sick to healthy. It is appropriate that it is decreasing. Again we use re(x+t) where x is the age of the life and of course it does not have to be of this form.

```
> re := t -> 2*(1+t/50)^(-1);
```

$$re := t \rightarrow \frac{2}{1+\frac{1}{50}\,t}$$

The force of interest is 0.05. In order to write down the equations we first need to calculate the premium. We will proceed as follows. Let us first pretend the policy is funded by a single premium payable upfront. That would be wh(0) in the solution of

```
> dsys1 :=
  {diff(wh(t),t)=(m(35+t)+si(35+t)+0.05)*wh(t
  )-si(35+t)*ws(t),
  diff(ws(t),t)=(ms(35+t)+re(35+t)+0.05)*ws(t
  )-20000-re(35+t)*wh(t), wh(30)=0,
  ws(30)=0};
```

$$dsys1 := \Big\{ \frac{d}{dt}\,\mathrm{wh}(t) = (0.2205000000$$

$$+\,0.00007585775\;10^{(1.330+0.038\,t)} + 0.002000000000\,t)\,\mathrm{wh}(t)$$

$$-\,(0.1700000000 + 0.002000000000\,t)\,\mathrm{ws}(t),\ \frac{d}{dt}\,\mathrm{ws}(t) =$$

$$\left(0.05095 + 0.000113786625\;10^{(1.520+0.038\,t)} + \frac{2}{\frac{17}{10}+\frac{t}{50}}\right)\mathrm{ws}(t)$$

$$-20000-\frac{2\,\mathrm{wh}(t)}{\frac{17}{10}+\frac{t}{50}},\ \mathrm{wh}(30)=0,\ \mathrm{ws}(30)=0\Big\}$$

```
> dsol1 := dsolve(dsys1, numeric, range =
  0..30);
```

$$dsol1 := \mathbf{proc}(x\_rkf45)\ \ldots\ \mathbf{end\ proc}$$

```
> dsol1(30);
```

$$[\,t=30.,\ \mathrm{wh}(t)=0.,\ \mathrm{ws}(t)=0.\,]$$

```
> dsol1(0);
```

$$[\,t=0.,\ \mathrm{wh}(t)=42794.8333026865148,$$
$$\mathrm{ws}(t)=57067.8796379095293\,]$$

Let us now calculate in a similar way the value of an annuity of 1 payable continuously while the life is healthy. This would be wh(0) in the solution of

```
> dsys2 :=
  {diff(wh(t),t)=(m(35+t)+si(35+t)+0.05)*wh(t
  )-si(35+t)*ws(t)-1,
  diff(ws(t),t)=(ms(35+t)+re(35+t)+0.05)*ws(t
  )-re(35+t)*wh(t), wh(30)=0, ws(30)=0};
```

$$dsys2 := \Big\{\mathrm{wh}(30)=0,\ \mathrm{ws}(30)=0,\ \frac{d}{dt}\,\mathrm{wh}(t)=(0.2205000000$$
$$+0.00007585775\,10^{(1.330+0.038\,t)}+0.002000000000\,t)\,\mathrm{wh}(t)$$

$$- (0.1700000000 + 0.002000000000\, t)\, \mathrm{ws}(t) - 1, \frac{d}{dt}\mathrm{ws}(t) =$$

$$\left(0.05095 + 0.000113786625\, 10^{(1.520 + 0.038\, t)} + \frac{2}{\frac{17}{10} + \frac{t}{50}}\right) \mathrm{ws}(t)$$

$$\left. - \frac{2\, \mathrm{wh}(t)}{\frac{17}{10} + \frac{t}{50}} \right\}$$

```
> dsol2 := dsolve(dsys2, numeric, range =
  0..30);
```

$$dsol2 := \mathbf{proc}(x\_rkf45) \;\; \ldots \;\; \mathbf{end\ proc}$$

```
> dsol2(0);
```

$$[t = 0., \mathrm{wh}(t) = 12.4724100560195588,$$
$$\mathrm{ws}(t) = 11.7303723907940558]$$

And so the continuous premium is

```
> evalf(42794.8333026865148/12.4724100560195588);
```

$$3431.159904$$

Lei us now solve Thiele's equations.

```
> dsys3 :=
  {diff(vh(t),t)=(m(35+t)+si(35+t)+0.05)*vh(t
  )-si(35+t)*vs(t)+3431.159904,
  diff(vs(t),t)=(ms(35+t)+re(35+t)+0.05)*vs(t
  )-re(35+t)*vh(t)-20000, vh(30)=0,
  vs(30)=0};
```

$$dsys3 := \Big\{ \mathrm{vh}(30) = 0,\ \mathrm{vs}(30) = 0,\ \frac{d}{dt}\mathrm{vh}(t) = (0.2205000000 + 0.00007585775\ 10^{(1.330 + 0.038\,t)} + 0.002000000000\,t)\,\mathrm{vh}(t) - (0.1700000000 + 0.002000000000\,t)\,\mathrm{vs}(t) + 3431.159904,$$
$$\frac{d}{dt}\mathrm{vs}(t) = \left(0.05095 + 0.000113786625\ 10^{(1.520 + 0.038\,t)} + \frac{2}{\frac{17}{10} + \frac{t}{50}}\right)\mathrm{vs}(t) - \frac{2\,\mathrm{vh}(t)}{\frac{17}{10} + \frac{t}{50}} - 20000 \Big\}$$

```
> dsol3 := dsolve(dsys3, numeric, range = 0..30);
```

$$dsol3 := \mathbf{proc}(x\_rkf45)\ \ldots\ \mathbf{end\ proc}$$

```
> dsol3(0);
```

$$[t = 0.,\ \mathrm{vh}(t) = -0.000248198026326917900,\ \mathrm{vs}(t) = 16819.0967395445478]$$

```
>
```

Note that vh(30) is almost 0 as it should be. What does vs(0) represent? Think.

We will also plot the two curves. We will discuss them in class. Can you spot a little problem and explain it? If interested open life4a.mws to see how to deal with it.

```
> with(plots);
>
Warning, the name changecoords has been redefined
```

[*animate*, *animate3d*, *animatecurve*, *arrow*, *changecoords*, *complexplot*, *complexplot3d*, *conformal*, *conformal3d*, *contourplot*, *contourplot3d*, *coordplot*, *coordplot3d*, *cylinderplot*, *densityplot*, *display*, *display3d*, *fieldplot*, *fieldplot3d*, *gradplot*, *gradplot3d*, *graphplot3d*, *implicitplot*, *implicitplot3d*, *inequal*, *interactive*, *listcontplot*, *listcontplot3d*, *listdensityplot*, *listplot*, *listplot3d*, *loglogplot*, *logplot*, *matrixplot*, *odeplot*, *pareto*, *plotcompare*, *pointplot*, *pointplot3d*, *polarplot*, *polygonplot*, *polygonplot3d*, *polyhedra_supported*, *polyhedraplot*, *replot*, *rootlocus*, *semilogplot*, *setoptions*, *setoptions3d*, *spacecurve*, *sparsematrixplot*, *sphereplot*, *surfdata*, *textplot*, *textplot3d*, *tubeplot*]

```
> odeplot(dsol3);
```




```
> odeplot(dsol3, [t,vs(t)]);
```



vs
20000
15000
10000
5000
5
10
15
20
25
30
t