In this worksheet we are going to discuss a contact that involves two types of sickness. A different benefit is paid in each case.
This is the force of mortality for healthy lives

```
> m := t -> 0.0005+0.00007585775*10^(0.038*t);
```

$$m := t \to 0.0005 + 0.00007585775\ 10^{(0.038\ t)} \tag{1}$$

We now define the mortality force for sick lives (both kinds have the same)

```
> ms:= t -> 0.00012+m(t+5);
```

$$ms := t \to 0.00012 + m(t+5) \tag{2}$$

This is the force of transition from the healthy state to the type A sick state.

```
> sa := t -> 0.01*(1+t/50);
```

$$sa := t \to 0.01\left(1 + \frac{1}{50}\,t\right) \tag{3}$$

This is the force of transition from the healthy state to the type B sick state.

```
> sb := t -> 0.02*(1+t/100);
```

$$sb := t \to 0.02\left(1 + \frac{1}{100}\,t\right) \tag{4}$$

This is the force of transition from the type A sick state to the healthy state.

```
> ra := t -> 3*(1+t/50)^(-1);
```

$$ra := t \to \frac{3}{1 + \frac{1}{50}\,t} \tag{5}$$

This is the force of transition from the type B sick state to the healthy state.

```
> rb := t -> 2*(1+t/30)^(-1);
```

$$rb := t \to \frac{2}{1 + \frac{1}{30}\,t} \tag{6}$$

The benefit is 20000 per annum while sick due to A and 40000 while sick due to B. We start by calculating the expected valur of the benefit (force of interest is 0.05). The policy is for 35 years and the policyholder is 30 years old. Consider the following system of equations

```
> dsys1 := {diff(wh(t),t)=(m(30+t)+sa(30+t)+sb(30+t)+0.05)*wh(t)-sa
  (30+t)*wa(t)-sb(30+t)*wb(t), diff(wa(t),t)=(ms(30+t)+ra(30+t)
  +0.05)*wa(t)-ra(30+t)*wh(t)-20000,diff(wb(t),t)=(ms(30+t)+rb(30+t)
  +0.05)*wb(t)-rb(30+t)*wh(t)-40000, wh(35)=0, wa(35)=0, wb(35)=0};
```

$$dsys1 := \Big\{ wh(35) = 0,\ wa(35) = 0,\ wb(35) = 0,\ \frac{d}{dt}\,wh(t) = \Big(0.09250000000 \tag{7}$$

$$+ 0.00007585775\ 10^{(1.140 + 0.038\ t)} + 0.0004000000000\ t\Big)\,wh(t) - (0.01600000000 + 0.0002000000000\ t)\,wa(t) - (0.02600000000 + 0.0002000000000\ t)\,wb(t),\ \frac{d}{dt}\,wa(t) = \Bigg(0.05062 + 0.00007585775\ 10^{(1.330 + 0.038\ t)}$$

$$+ \frac{3}{\frac{8}{5} + \frac{1}{50}\,t}\Bigg)\,wa(t) - \frac{3\,wh(t)}{\frac{8}{5} + \frac{1}{50}\,t} - 20000,\ \frac{d}{dt}\,wb(t) = \Bigg(0.05062$$

$$+ 0.00007585775\ 10^{(1.330 + 0.038\, t)} + \frac{2}{2 + \frac{1}{30}\, t}\Bigg)\, wb(t) - \frac{2\, wh(t)}{2 + \frac{1}{30}\, t} - 40000 \Bigg\}$$

```
> dsol1 := dsolve(dsys1, numeric, range = 0..35);
```

$$dsol1 := \mathbf{proc}(x\_rkf45)\ \ldots\mathbf{end\ proc} \tag{8}$$

```
> dsol1(35);
```

$$[t = 35., wa(t) = 0., wb(t) = 0., wh(t) = 0.] \tag{9}$$

```
> dsol1(0);
```

$$[t = 0., wa(t) = 32467.8697750551000, wb(t) = 60087.3166875669849, wh(t) = 22631.1911587924806] \tag{10}$$

```
>  dsol1(10);
```

$$[t = 10., wa(t) = 32536.5792010266850, wb(t) = 64572.8408236166724, wh(t) = 21690.1943359052494] \tag{11}$$

```
> dsol1(20);
```

$$[t = 20., wa(t) = 28908.8457970307754, wb(t) = 64955.9706630295258, wh(t) = 17140.1775952215139] \tag{12}$$

We now calculate the value of an annuity of 1 payable while healthy

```
> dsys2 := {diff(vh(t),t)=(m(30+t)+sa(30+t)+sb(30+t)+0.05)*vh(t)-sa
  (30+t)*va(t)-sb(30+t)*vb(t)-1, diff(va(t),t)=(ms(30+t)+ra(30+t)
  +0.05)*va(t)-ra(30+t)*vh(t),diff(vb(t),t)=(ms(30+t)+rb(30+t)+0.05)
  *vb(t)-rb(30+t)*vh(t), vh(35)=0, va(35)=0, vb(35)=0};
```

$$dsys2 := \Bigg\{ vh(35) = 0, va(35) = 0, vb(35) = 0, \frac{d}{dt}\, vh(t) = \big(0.09250000000 + 0.00007585775\ 10^{(1.140 + 0.038\, t)} + 0.0004000000000\, t\big)\, vh(t) - (0.01600000000 + 0.0002000000000\, t)\, va(t) - (0.02600000000 + 0.0002000000000\, t)\, vb(t) - 1, \frac{d}{dt}\, va(t) = \Bigg(0.05062 + 0.00007585775\ 10^{(1.330 + 0.038\, t)} + \frac{3}{\frac{8}{5} + \frac{1}{50}\, t}\Bigg)\, va(t) - \frac{3\, vh(t)}{\frac{8}{5} + \frac{1}{50}\, t}, \frac{d}{dt}\, vb(t) = \Bigg(0.05062 + 0.00007585775\ 10^{(1.330 + 0.038\, t)} + \frac{2}{2 + \frac{1}{30}\, t}\Bigg)\, vb(t) - \frac{2\, vh(t)}{2 + \frac{1}{30}\, t} \Bigg\} \tag{13}$$

```
> dsol2 := dsolve(dsys2, numeric, range = 0..35);
```

$$dsol2 := \mathbf{proc}(x\_rkf45)\ \ldots\mathbf{end\ proc} \tag{14}$$

```
> dsol2(35);
>
```

$$[t = 35., va(t) = 0., vb(t) = 0., vh(t) = 0.] \tag{15}$$

```
> dsol2(0);
```

$$[t = 0., va(t) = 14.6184156790790798, vb(t) = 14.1857384321353735, vh(\tag{16}$$

$t) = 15.1289476700398602]$

```
> dsol2(10);
```

$[t=10., va(t)=12.2530610857598834, vb(t)=11.7359916842049206, vh(t)=12.8251282078342152]$ (17)

```
> dsol2(20);
```

$[t=20., va(t)=8.75003786876911604, vb(t)=8.15449655818818542, vh(t)=9.38374843565046922]$ (18)

We will now calculate the continuous premium

```
> evalf(22631.1911587924806/15.1289476700398602);
```

$$1495.886671 \tag{19}$$

```
>
```

We will now calculate reserves at various times

```
> dsys3 := {diff(uh(t),t)=(m(30+t)+sa(30+t)+sb(30+t)+0.05)*uh(t)-sa(30+t)*ua(t)-sb(30+t)*ub(t)+1495.886671, diff(ua(t),t)=(ms(30+t)+ra(30+t)+0.05)*ua(t)-ra(30+t)*uh(t)-20000,diff(ub(t),t)=(ms(30+t)+rb(30+t)+0.05)*ub(t)-rb(30+t)*uh(t)-40000, uh(35)=0, ua(35)=0, ub(35)=0};
>
```

$$dsys3 := \Bigg\{ uh(35)=0, ua(35)=0, ub(35)=0, \frac{d}{dt}\, uh(t) = \big(0.09250000000 + 0.00007585775\; 10^{(1.140\,+\,0.038\,t)} + 0.0004000000000\,t\big)\, uh(t) - (0.01600000000 + 0.0002000000000\,t)\, ua(t) - (0.02600000000 + 0.0002000000000\,t)\, ub(t) + 1495.886671, \frac{d}{dt}\, ua(t) = \left(0.05062 + 0.00007585775\; 10^{(1.330\,+\,0.038\,t)} + \frac{3}{\frac{8}{5}+\frac{1}{50}\,t} \right) ua(t) - \frac{3\, uh(t)}{\frac{8}{5}+\frac{1}{50}\,t} - 20000, \frac{d}{dt}\, ub(t) = \left(0.05062 + 0.00007585775\; 10^{(1.330\,+\,0.038\,t)} + \frac{2}{2+\frac{1}{30}\,t} \right) ub(t) - \frac{2\, uh(t)}{2+\frac{1}{30}\,t} - 40000 \Bigg\} \tag{20}$$

```
> dsol3 := dsolve(dsys3, numeric, range = 0..35);
```

$$dsol3 := \textbf{proc}\,(x\_rkf45)\ \ldots\textbf{end proc} \tag{21}$$

```
> dsol3(0);
```

$\left[t=0., ua(t)=10600.3773989651164, ub(t)=38867.0588520231758, uh(t)=8.68163112954789542\; 10^{-7}\right]$ (22)

```
> dsol3(35);
>
```

$$[t=35., ua(t)=0., ub(t)=0., uh(t)=0.] \tag{23}$$

```
> dsol3(20);
```

$[t=20., ua(t)=15819.7762495145180, ub(t)=52757.7643944067458, uh(t)=3103.15356926179766]$ (24)

```
> dsol3(10);
```
$$[t = 10., ua(t) = 14207.3865076616985, ub(t) = 47017.1211510149151, uh(t) = 2505.25623561786369] \quad (25)$$

```
> dsol3(30);
```
$$[t = 30., ua(t) = 14189.5722474873473, ub(t) = 53123.2583605540131, uh(t) = 534.764209980887018] \quad (26)$$

```
> dsol3(32);
```
$$[t = 32., ua(t) = 13439.8220808924707, ub(t) = 48385.7061707270114, uh(t) = -373.163400425207158] \quad (27)$$

We see that the problem with negative reserves is here again. It can be dealt with as in the previous worksheet.